Newsgroup: it.comp.lang.visual-basic
Titolo: [?] Icone nei SubItems di ListView

Ciao a tutti.
Vorrei inserire in alcune colonne di un Listview solo delle
icone, senza testo.   Purtroppo pero' sembra che nei
subitems dei ListItem sia possibile inserire solo testo.
Eppure in Outlook Express, ad esempio, questo effetto
esiste. Forse non sara' un Listview standard, ma chissa'...
magari con qualche messaggio al listview si puo' fare...
Grazie
Bepi

 scrisse:

Prova a cambiare la proprieta' View della listvies in lvwIcon!

> Prova a cambiare la proprieta' View della listvies in lvwIcon!
A me serve in modalita' report.   Una decina di campi text,
tre o quattro campi flag da riempire con iconcine...
Ho fatto l'esempio di Outlook Express non a caso...
Ciao
Bepi

> Prova a cambiare la proprieta' View della listvies in lvwIcon!

Ops... Non avevo capito...
Ho trovato questo codice, ma lascio a te la prova (sorry, il lavoro
preme), dovrebbe fare quello che dici tu...
--------------------------
Icons in SubItem columns
To begin, drop a ListView control and an ImageList control 
onto a form. Add two 16x16 images to the ImageList.  
Set the ListView to display itself in report style, add two 
column headers, then bind the ImageList control to the 
ListView as the small icon image list.
Option Explicit
Private Type LVITEM
    Mask        As Long
    Item        As Long
    SubItem     As Long
    State       As Long
    StateMask   As Long
    Text        As String
    TextMax     As Long
    Image       As Long
    Param       As Long
    Indent      As Long
End Type
Private Const LVM_FIRST = &H1000&
Private Const LVM_SETITEM = (LVM_FIRST + 6)
Private Const LVM_SETEXTENDEDLISTVIEWSTYLE = (LVM_FIRST + 54)
Private Const LVM_GETEXTENDEDLISTVIEWSTYLE = (LVM_FIRST + 55)
Private Const LVS_EX_SUBITEMIMAGES = &H2&
Private Const LVIF_IMAGE = &H2&
Private Declare Function SendMessage Lib "user32" Alias "SendMessageA"
_
    (ByVal hwnd As Long, ByVal wMsg As Long, ByVal wParam As Long, _
    lParam As Any) As Long
'_________________________________________________________
Private Sub Form_Load()
    Dim LV              As LVITEM
    Dim lStyle          As Long
    Dim li              As ListItem
 
    '   Add a ListItem
    Set li = ListView1.ListItems.Add(, , "Item 1", , 1)
    li.SubItems(1) = "SubItem 1"
 
    '   Set the ListView style to show sub-item images
    lStyle = SendMessage(ListView1.hwnd, _
                LVM_GETEXTENDEDLISTVIEWSTYLE, 0, ByVal 0&)
    lStyle = lStyle Or LVS_EX_SUBITEMIMAGES
    SendMessage ListView1.hwnd,  _
		LVM_SETEXTENDEDLISTVIEWSTYLE, 0, ByVal lStyle
 
    '   Change the ListItem's sub-item image
    With LV
        .Mask = LVIF_IMAGE
        .Item = li.Index - 1
        .SubItem = 1        '1-based index of sub-item
        .Image = 1          '0-based index of list image
    End With
    SendMessage ListView1.hwnd, LVM_SETITEM, li.Index - 1, LV
End Sub
'When run, this code will display a single ListItem with the first
image as
'the Item 's icon and the second image as the sub-item's icon.

...ecco, ora forse ci siamo, grazie.   Non ho ancora provato
ma lo faccio subito.   Grazie
Ciao
Bepi

Bepi ha scritto:

Prova a guardare le proprietą
ListView.ListItems(Index).Icon
ListView.ListItems(Index).SmallIcon
lvwMessaggi.ListItems(Index).ListSubItems(SubIndex).ReportIcon
-- 
Samuele Battarra
prova il mio newsreader:
http://digilander.iol.it/battarsa/programmi/news/

 Prova a guardare le proprietà



Funziona, grazie mille ! (e funziona anche con le
listview del SP precedente)
Grazie
Bepi

 ha scritto nel messaggio

Prego! Buon lavoro!